

reduce the number after BL=. If you want to sit back and let the ball bounce on it's own then simply increase the bat size until it fills the bottom of the screen.

The Program

```

10 REM Breakout
20 LET HS=0
22 LET BL=7
25 LET BA$="": FOR L=1 TO BL: LET BA$=BA
  $+CHR$(132): NEXT L: LET BA$=" "+BA$+" "
27 LET BE=31-BL
30 GOSUB 2000
49 REM Main new game loop
50 LET SC=0
55 REM Main new frame loop
57 LET BH=90
60 GOSUB 1000
70 LET BX=10: GOSUB 3000
80 LET X=5: LET Y=20
85 LET CX=1: LET CY=-1
90 LET GF=1
99 REM Main bounce and Score loop
100 GOSUB 4000
105 IF GF=0 THEN GOTO 8000
110 GOSUB 5000
120 GOSUB 6000
130 COLOUR 1,15
140 CSR 7,0: PRINT SC;
199 IF BH<>0 THEN GOTO 100
200 REM End of frame
210 FOR L=10 TO 250 STEP 5
220 SOUND 1,L,15: SOUND 2,L+6,15
225 PAUSE 30
230 NEXT L
235 SOUND 1,0,0: SOUND 2,0,0
240 CLS : CSR 3,10
250 PRINT "Your score is now :";SC
260 IF SC>HS THEN PRINT "You have the h
  igh score": PRINT : PRINT "and you're st
  ill playing": GOTO 280
270 PRINT "You are ";HS-SC;" points behi
  nd": PRINT : PRINT " the high score"

```

```

280 PAUSE 4000
290 GOTO 55
1000 REM Draw screen layout
1010 VS 4: COLOUR 4,1: COLOUR 0,1: COLOU
  R 2,1: CLS
1020 COLOUR 1,13: FOR L=1 TO 21
1025 CSR 0,L: PRINT CHR$(129): CSR 31,L:
  PRINT CHR$(130);: NEXT L
1030 CSR 1,1: FOR L=1 TO 30: PRINT CHR$(
  134);: NEXT L
1040 FOR L=1 TO 30: COLOUR 1,3: CSR L,5:
  PRINT CHR$(133);: CSR L,12: PRINT CHR$(
  133);
1045 COLOUR 1,6: CSR L,8: PRINT CHR$(133
  );: NEXT L
1050 CSR 1,0: COLOUR 1,4: PRINT "SCORE:
  HIGH: ";HS
1099 RETURN
2000 REM Generate the Patterns
2010 GENPAT 1,129,4,2,1,2,4,2,1,2
2020 GENPAT 1,130,32,64,128,64,32,64,128
  ,64
2030 GENPAT 1,131,24,62,126,255,255,126,
  62,24
2040 GENPAT 1,132,255,255,170,85,72,16,0
  ,0
2050 GENPAT 1,133,255,170,85,170,85,170,
  85,255
2060 GENPAT 1,134,0,0,0,0,0,0,0,255
2099 RETURN
3000 REM Draw bat on screen
3010 CSR BX-1,22
3020 COLOUR 1,9
3030 PRINT BA$;
3099 RETURN
4000 REM set up new ball position
4001 REM change score if appropriate
4010 LET C=X+CX: LET D=Y+CY
4020 IF C<1 OR C>30 THEN LET CX=-CX: LE
  T SC=SC+2: GOTO 4010
4030 IF D<2 THEN LET CY=-CY: LET SC=SC+
  4: LET CX=INT(RND*3)-1: GOTO 4010
4035 IF D>22 THEN LET GF=0: RETURN
4040 CSR C,D: LET E=ASC(SPK$)
4050 IF E=32 THEN RETURN
4055 IF E=132 THEN GOTO 4080

```



```
4057 LET BH=BH-1
4060 LET SC=SC+D: PRINT CHR$(8); " "
4070 LET CX=INT(RND*3)-1: LET CY=-CY: GO
TO 4010
4080 LET SC=SC+20: LET CY=-CY
4085 LET CX=INT(RND*3)-1
4090 GOTO 4010
4099 RETURN
5000 REM erase old ball, plot new ball
5010 CSR C,D: PRINT CHR$(131): CSR X,Y:
PRINT " "
5020 LET X=C: LET Y=D
5099 RETURN
6000 REM move bat if necessary
6010 LET A=ASC(INKEY$)
6020 IF A=8 AND BX>1 THEN LET BX=BX-1:
GOSUB 3000
6030 IF A=25 AND BX<BE THEN LET BX=BX+1
: GOSUB 3000
6099 RETURN
8000 REM Player has lost the ball
8010 CLS : CSR 0,10: PRINT "Game Over"
8020 CSR 0,14: PRINT "You scored :";SC
8030 IF SC>HS THEN CSR 0,17: PRINT "Whi
ch is the high score": LET HS=SC: GOTO
8050
8040 CSR 0,17: PRINT "Slightly (";HS-SC;
") less than": CSR 0,19: PRINT "the high
score"
8050 CSR 0,20: PRINT "Do you want to pla
y again": CSR 14,22: PRINT "(Y/N)"
8060 LET I$=INKEY$
8070 IF I$<>"Y" AND I$<>"N" THEN GOTO 8
060
8080 IF I$="Y" THEN GOTO 50
8090 VS 5: CLS : CSR 10,10: PRINT "Type
RUN to restart"
8099 STOP
```